
A Succinct Formalization of the Completeness of

Abstract
I succinctly formalize the soundness and completeness of a small Hilbert system for first-order
logic in the proof assistant Isabelle/HOL. The proof combines and details ideas from de Bruijn,
Henkin, Herbrand, Hilbert, Hintikka, Lindenbaum, Smullyan and others in a novel way, and I use a
declarative style, custom notation and proof automation to obtain a readable formalization. The
formalized definitions of Hintikka sets and Herbrand structures allow open and closed formulas to
be treated uniformly, making free variables a non-concern. This paper collects important techniques
in mathematical logic in a way suited for both study and further work.

1 Introduction

The completeness of first-order logic has been known since Gödel’s work in 1929 [19]. Proof
assistants too have a long history [18], with de Bruijn initiating the Automath project in 1967
and the first system of LCF, an Isabelle/HOL predecessor, being developed in 1972. Despite
of this, I am unaware of a formalization of completeness in a proof assistant with a focus on
explaining the core techniques. The ideas involved in such a proof deserve to be documented
and detailed in a formalization where the proof assistant guarantees precision and correctness.
This effort benefits students trying to understand mathematical logic and researchers looking
for a base for their own work. I start from a Hilbert system, partly because I am unaware of
a formalization which does the same, and partly because its simplicity allows me to focus
on the ideas of the completeness proof itself. While other deduction systems may be more
popular for first-order logic, Hilbert systems are still prominent in areas like modal logic.

This paper builds especially on work by Smullyan [40] and Henkin [21]. The Hilbert
system of choice is Smullyan’s System Q1 [40, p. 81] and the completeness proof resembles
the “more direct construction” near the end of his book [40, p. 96] (a construction that
was pointed out to him by Henkin himself). This paper formalizes ideas by de Bruijn,
Henkin, Herbrand [23], Hilbert, Hintikka, Lindenbaum and Smullyan in an attempt to give a
“strikingly direct” [40, p. 96] completeness proof formalized in a modern proof assistant.

Smullyan includes a generalization rule for the universal quantifier that works under an
assumption (i.e. to the right of an implication) rather than on a standalone formula. This
extra generality makes it very suited for my purposes, where I always work under a number

8:2 A Succinct Formalization of the Completeness of First-Order Logic

of assumptions. Smullyan does not consider function symbols as part of his syntax, but his
System Q1 is easily extended to cover these: simply allow for any term in the instantiation
axiom. Barwise [1] makes the same modification.

I use the proof assistant Isabelle/HOL [34]. Isabelle is a generic proof assistant and
Isabelle/HOL is the version based on higher-order logic. This paper includes a number of
Isabelle listings, all taken from the formalization after it has been exported to LATEX. These
listings appear either in bulleted lists or prefixed by an Isabelle command in bold and should
therefore be clearly distinguishable from the surrounding text. Any such listing has been
checked and verified by the proof assistant. I sometimes use these listings to explain proofs.
In these cases, I do not include the commands that justify each step of reasoning, but only the
intermediate statements themselves. For clarity, I have omitted many types from the main
text. Some of these can be found in Table 1 on page 6. The full formalization (under 700
lines) is available in the Archive of Formal Proofs [17], which referees Isabelle formalizations
and, when accepted, keeps them up to date with the latest version of the proof assistant.

Contributions

The main contribution of this paper is a succinct formalization of the definitions and proofs
that make up the synthetic style, a widely applicable method of proving completeness.

As a smaller contribution, this is, to my knowledge, the first formalization of completeness
for classical first-order logic that starts from a Hilbert system. However, several formalizations
that start from other proof systems are available (cf. Section 2) and the relations between
various proof systems have also been formalized, see for instance recent work by Laurent [27]
in Coq on translating between Hilbert systems and natural deduction for first-order logic.

On the more technical side, I formalize a Herbrand universe which, like in Herbelin
and Ilik’s [22] unformalized proof, consists of all terms, not just those without variables.
Combined with a Hintikka set in the style of Forster et al. [11] in Coq, based on the absence
of formulas rather than the presence of their negations, but which, unlike theirs, contains
open formulas as well as closed, I naturally formalize completeness for all valid formulas.

Isabelle/HOL Overview

This section seeks to give a quick overview of the Isabelle/HOL features used later. Nipkow
and Klein [33, Part 1] give a more complete introduction.

The higher-order logic of Isabelle/HOL can be reasonably understood as functional
programming + logic [33]. The datatype command defines a new type from a series of
constructors, where each can be given custom syntax. The natural numbers are built from
the nullary constructor 0 and unary Suc. The constructors True and False belong to the
built-in type bool. The usual connectives and quantifiers from first-order logic (−→, ∀ ,
etc.) are available for bool, as well as if-then-else expressions. We can write total functions
over datatypes using primrec for primitive recursive functions and fun for more advanced
definitions. The type constructor A ⇒ B denotes a function from A to B. Instead of concrete
types, we can also use type variables ′a, ′b, etc. These stand in the place of other types. The
term UNIV stands for the set of all values of a given type.

Another important built-in type is ′a list, the type of lists whose elements are of type ′a.
These are built from [], the empty list, and #, an infix constructor that adjoins an element
to an existing list. The notation [a, b, c] is shorthand for these primitive operations. The
function set turns a list into a set of its elements. The higher-order function map applies a
given function to every element of a list.

Function application resembles functional programming languages in that f(x, y) is
written as f x y. The function f(x := y) maps x to y and every other element x′ to f x′.
Anonymous functions can be built using λ-expressions, e.g. λn. n + n for f(n) = n + n.

In proofs, the meta-logical implication =⇒ separates assumptions from conclusions. These
can also be distinguished using the assumes and shows keywords, using and as a separator
when there are multiple assumptions or conclusions. The keyword have states an intermediate
fact in a proof and the keywords then, moreover and ultimately bind these statements
together in different ways. The keyword let introduces a local abbreviation and obtain
eliminates an existential statement; for quantifies a statement universally.

The command definition introduces a new definition that is hidden behind a name
and must be explicitly unfolded, while an abbreviation is unfolded by the parser. The
inductive command also makes use of the meta-logical implication. This command allows
us to specify the least predicate closed under some given rules. For instance a predicate that
denotes whether a formula can be derived, specified by axioms and inference rules. A locale
in Isabelle, as used here, names an association between terms and assumptions about them.
We could, for instance, specify a group as a set and a binary operation coupled with the
group axioms. To instantiate the locale we must give concrete terms and show that they
satisfy the assumptions. When assuming a locale, we assume the conditions hold for the
terms.

The axiom of choice is available as Hilbert’s choice operator: the expression SOME x. P x
returns some element x that satisfies the predicate P, when such an element exists.

Overview of Paper

I discuss related work next (Section 2). In Section 3, I formalize the syntax of first-order
logic in Isabelle/HOL, including the idea of de Bruijn indices. This idea is developed
further in Section 4 on the semantics of terms and formulas. Section 5 formalizes the proof
system and its soundness, including all the operations necessary to do so. This includes the
instantiation of universal quantifiers, propositional tautologies and a range of lemmas. I
prove the completeness in Section 6 where I introduce the notion of a maximal consistent set,
the Lindenbaum construction and the model existence theorem for Hintikka sets. I discuss
challenges and choices in Section 7 and conclude with thoughts on future work in Section 8.

2 Related Work

The completeness of first-order logic itself has a long history, starting with Gödel’s proof [19]
and Henkin’s later simplification [21]. Smullyan [40], Barwise [1] and Fitting [10] all include
completeness proofs in their texts. Smullyan’s main completeness proof is an “analytic” proof
for a tableau system. He devotes only two pages to the “synthetic” (also called Henkin-style)
completeness proof of System Q1 [40, pp. 96–97] that I formalize a version of here. Barwise [1]
considers System Q1 extended with axioms for equality and gives a quite different proof
that relies on a reduction to propositional logic (and the completeness of propositional logic).
Fitting [10] proves completeness for tableaux and resolution similarly to Smullyan and leaves
the completeness of his Hilbert system as an exercise for the reader. This paper spells out
the synthetic completeness proof for first-order logic, starting from a Hilbert system rather
than from tableaux, resolution or similar.

The synthetic style has been successfully applied in several formalizations lately. From [13]
used it to formalize the completeness of a Hilbert system for propositional logic in Isa-
belle/HOL. Berghofer [3] formalized natural deduction for first-order logic in Isabelle/HOL

TYPES 2021



8:4 A Succinct Formalization of the Completeness of First-Order Logic

following the work by Fitting [10]. My formalization of the syntax and semantics of first-order
logic and the Lindenbaum construction is inspired by his work. My formalization of Hintikka
sets and proof of the model existence theorem, however, differ from his and is inspired by
Herbelin and Ilik [22] and Forster et al. [11]. In particular, I prove completeness for open and
closed formulas together, where Berghofer’s completeness proof only covers closed formulas
and is extended to cover open formulas afterwards. From, Schlichtkrull and Villadsen [14, 16]
built on Berghofer’s work to formalize the completeness of both a sequent calculus and
tableau system for first-order logic. They also described Berghofer’s formalization in detail.
Bentzen [2] formalized the completeness of a Hilbert system for the modal logic S5 in Lean.
Jørgensen et al. [26] gave a synthetic completeness proof for a tableau system for basic hybrid
logic, which From [12, 15] formalized in Isabelle/HOL.

I am far from the first to formalize the completeness of first-order logic, but my formaliz-
ation is the only one that proves completeness for a Hilbert system for classical first-order
logic. Shankar [39] formalized a tautology checker for first-order logic in the Boyer-Moore
theorem prover, but notably did not formalize first-order completeness. Harrison [20] also
formalized first-order logic in higher-order logic (but HOL rather than Isabelle/HOL). He
did not formalize a proof system but rather model-theoretic results like compactness and
the Löwenheim-Skolem theorem. Margetson and Ridge [29] formalized the completeness
of first-order logic without functions in Isabelle/HOL via a sequent calculus. Braselmann
and Koepke [7] did the same in their Mizar formalization. Schlichtkrull [37, 38] formalized
the completeness of first-order resolution in Isabelle/HOL. Michaelis and Nipkow [30, 31]
did not formalize first-order logic, but did formalize a Hilbert system for propositional
logic in Isabelle/HOL. They proved completeness via translation from a sequent calculus
with an analytic completeness proof. Blanchette, Popescu and Traytel [5, 6] formalized
analytic completeness of abstract sequent calculus and tableau systems for first-order logic in
Isabelle/HOL. Blanchette [4] outlines formalizations of logical meta-theory in Isabelle/HOL.

The completeness proof presented here relies on Lindenbaum’s lemma: that any consistent
set of formulas has a maximal consistent extension. The proof is non-constructive since, for
the given semantics, Lindenbaum’s lemma is equivalent to Weak König’s Lemma [22, 24].
There are, however, a number of formalizations of completeness in other meta-theories (and
necessarily using other semantics). Veldman [43] gave an intuitionistic completeness theorem
for intuitionistic predicate logic in 1976. Persson [35] formalized the completeness of sequent
calculus and natural deduction for intuitionistic first-order logic in the ALF proof assistant,
but only defined a Hilbert system without further proof. His models are based on formal
topology. Constable and Bickford [8] constructively proved completeness for intuitionistic
first-order logic in the proof assistant Nuprl. Ilik [25] formalized multiple constructive proofs
of first-order completeness in the proof assistant Coq using novel variants of Kripke models
for full classical and intuitionistic first-order logic. Forster et al. [11] recently analyzed
the computational content of completeness theorems for various semantics and for natural
deduction and sequent calculus systems. They mechanized their results in constructive type
theory using Coq.

3 Syntax

The following syntax will be our starting point.
A term t is either a variable x or a function symbol f applied to a number of other terms:

s, t ::= x | f(t1, . . . , tn)

A function symbol applied to zero terms is called a constant and is denoted by a.

A formula p is either falsity (denoted ⊥), a predicate symbol P applied to a list of terms,
an implication (−→) between two formulas or a universally quantified formula:

p, q ::= ⊥ | P (t1, . . . , tn) | p −→ q | ∀x. p(x)

I apply a number of techniques to make this syntax suitable for formalization. First,
I represent the variables with de Bruijn indices [9]. Instead of connecting quantifiers and
variables by using the same variable symbol x, each variable is a natural number n that is
understood to be connected to the nth quantifier, starting from the innermost. For example,
the formula ∀x. ∀y. P (x, y) is represented as ∀ ∀ P (1, 0). This makes it simpler to define
capture-avoiding instantiation, which we need for the proof system.

Second, to distinguish variables, function symbols and predicate symbols in the proof as-
sistant, I prefix each kind by a distinct symbol: † for function symbols, ‡ for predicate symbols
and # for variables. The formula ∀ P (f(0)) is then written (for now) as ∀ ‡P (†f(#0)).

Finally, lists of argument terms are represented as proper Isabelle/HOL lists, so the
term f(a) becomes †f [a].

The (parameterized) datatype ′f tm of terms with function symbols of type ′f becomes:

datatype (params-tm: ′f ) tm
= Var nat (#)
| Fun ′f ( ′f tm list) (†)

The annotation params-tm generates a function of that name from terms to ′f sets: it
collects all values of type ′f in a given term. I discuss its properties in Section 5.1.

The following abbreviates a constant, as I use these frequently:

abbreviation Const (⋆) where ⋆a ≡ †a []

The datatype ( ′f, ′p) fm of formulas with functions symbols of type ′f and predicate
symbols of type ′p becomes:

datatype (params-fm: ′f , ′p) fm
= Falsity (⊥)
| Pre ′p ( ′f tm list) (‡)
| Imp (( ′f , ′p) fm) (( ′f , ′p) fm) (infixr −→ 55 )
| Uni (( ′f , ′p) fm) (∀ )

The custom notation for each syntactic case is enclosed in parentheses (infixr specifies
right associativity and 55 specifies parsing priority). I use bold symbols to avoid conflicts
with existing syntax. The notation params-fm, similarly to for terms, generates a function
which produces a set of all function symbols in a given formula.

The Isabelle command term checks the type of an expression. Given the above definitions,
we can try our syntax, here with strings for the types of function and predicate symbols:

term ∀ (⊥ −→ ‡ ′′P ′′ [† ′′f ′′ [#0 ]])

In regular notation this would be ∀x. ⊥ −→ P (f(x)).
The following abbreviation for negation will ease notation: ¬ p ≡ p −→ ⊥ .
Similar notations can easily be introduced for conjunction, disjunction, the existential

quantifier etc. since in classical logic, these can be defined using the given syntax.
It should be noted that since arities are implicit in the datatypes above, we unfortunately

cannot represent finite signatures. The awarded benefit is that we do not need a predicate to
distinguish between correct and incorrect syntax.

TYPES 2021

Table 1 Type signatures for selected functions.

semantics-tm (nat ⇒ ′a) ⇒ ( ′f ⇒ ′a list ⇒ ′a) ⇒ ′f tm ⇒ ′a
semantics-fm (nat ⇒ ′a) ⇒ ( ′f ⇒ ′a list ⇒ ′a) ⇒ ( ′p ⇒ ′a list ⇒ bool) ⇒ ( ′f, ′p) fm ⇒ bool
shift (nat ⇒ ′a) ⇒ nat ⇒ ′a ⇒ nat ⇒ ′a
boolean ( ′p ⇒ ′f tm list ⇒ bool) ⇒ (( ′f, ′p) fm ⇒ bool) ⇒ ( ′f, ′p) fm ⇒ bool
Axiomatic ( ′f, ′p) fm ⇒ bool
imply ( ′f, ′p) fm list ⇒ ( ′f, ′p) fm ⇒ ( ′f, ′p) fm
consistent ( ′f, ′p) fm set ⇒ bool
extend ( ′f, ′p) fm set ⇒ (nat ⇒ ( ′f, ′p) fm) ⇒ nat ⇒ ( ′f, ′p) fm set
witness ′f set ⇒ ( ′f, ′p) fm ⇒ ( ′f, ′p) fm set
hmodel ( ′f, ′p) fm set ⇒ ( ′f, ′p) fm ⇒ bool

4 Semantics

The semantics of first-order logic has two parts: one for terms and one for formulas. I
formalize both as functions.

4.1 Terms
A term evaluates to an element of the domain. It does so under an environment (a mapping
from variables to the domain) and a function denotation (a mapping from function symbols
to functions on the domain).

In Isabelle, I represent the domain as a type (variable) and the environment as a
function E from the natural numbers (the de Bruijn indices) to that type. Similarly, the
function denotation becomes the function F from function symbols to functions on the
domain. This results in the following definition:

primrec semantics-tm ((|-, -|)) where
(|E , F |) (#n) = E n
| (|E , F |) (†f ts) = F f (map (|E , F |) ts)

The semantics of a variable is given by the environment and in the case of a function
application †f ts, we must first evaluate all the argument terms ts (using map) and then
apply the function denoted by f.

Here (|E, F |) denotes the function from terms to the domain, given by the environment E

and function denotation F . As seen in the clause above for functions, this notation lets me
conveniently “bundle” a given E and F so they can be applied to any term without the need
for anonymous functions. I use a similar notation [[E, F, G]] for the semantics of formulas.

4.2 Formulas
I use a deep embedding where formulas evaluate to a truth value under an environment E, a
function denotation F and a predicate denotation, dubbed G, that maps predicate symbols
to predicates on the domain. I formalize it as follows:

primrec semantics-fm ( [[-, -, -]] ) where
[[-, -, -]] ⊥ = False
| [[E , F , G]] (‡P ts) = G P (map (|E , F |) ts)
| [[E , F , G]] (p −→ q) = ([[E , F , G]] p −→ [[E , F , G]] q)
| [[E , F , G]] (∀ p) = (∀ x. [[E⟨0 :x⟩, F , G]] p)

The formula ⊥ is always False and the semantics of a predicate is similar to that of a
function application. For implication each subformula is evaluated to a truth value and the
connective is interpreted using the built-in implication. Similarly, I use the built-in universal
quantifier to interpret the object-level quantifier. The notation E⟨0:x⟩ is explained next.

4.3 Shifting
The expression E⟨n:x⟩ modifies the environment E such that variable n is assigned x, any
smaller variable m is assigned E m and any larger variable m is assigned E (m − 1). This
shift operation has the following definition:

definition shift ( -⟨-:-⟩) where
E⟨n:x⟩ m ≡ if m < n then E m else if m = n then x else E (m−1 )

To understand the shifting operation on larger variables, consider the following:

[[E, F, G]] (∀ ∀ ‡P [#0, #1])

By the semantics, this reduces to:

∀x. [[E⟨0:x⟩, F, G]] (∀ ‡P [#0, #1])

where the outer quantifier comes from the meta-logic. This again reduces to:

∀x. ∀y. [[E⟨0:x⟩⟨0:y⟩, F, G]] (‡P [#0, #1])

Thus, the terms are evaluated by (|E⟨0:x⟩⟨0:y⟩, F |). This is clearly correct for variable #0
since E⟨0:x⟩⟨0:y⟩ 0 = y as desired. We also want that #1 corresponds to the outer
meta-logic quantifier, namely that E⟨0:x⟩⟨0:y⟩ 1 = x. This is exactly what happens since
E⟨0:x⟩⟨0:y⟩ 1 = E⟨0:x⟩ (1 − 1) = x. Thus, the semantics reduces to the expected:

∀x. ∀y. G P [y, x]

Notice that any free variable in a formula (those with no corresponding quantifier) are not
affected by this shifting when it is coupled with the removal of an outer quantifier: they are
mapped to whatever E originally assigned them to. In this sense they behave like constants.

The following four lemmas will be used implicitly.

▶ Lemma 1 (Shifting). The first three results characterize the function and the last one
commutes a shift of variable 0 with another shift.

n = m =⇒ E⟨n:x⟩ m = x
m < n =⇒ E⟨n:x⟩ m = E m
n < m =⇒ E⟨n:x⟩ m = E (m−1 )
(E⟨n:y⟩⟨0 :x⟩) = (E⟨0 :x⟩⟨n+1 :y⟩)

Proof. Immediate from the definition. ◀

5 Proof System

To define the proof system I must first define a number of operations needed to state the
side conditions and transformations of formulas.

TYPES 2021



8:8 A Succinct Formalization of the Completeness of First-Order Logic

5.1 Parameters
The proof rule for the universal quantifier will generalize a fresh constant to a quantified
variable. To perform this freshness check, I use the functions params-tm and params-fm
that Isabelle generates automatically from the datatype declarations above. These collect
all function symbols in terms and formulas, respectively, and would also be easy to define
recursively. Similarly to Smullyan [40], I abbreviate function symbol to parameter.

The following definition generalizes params-fm to a set of formulas:

abbreviation params S ≡
⋃

p ∈ S . params-fm p

I need a few lemmas about parameters for later.

▶ Lemma 2 (Finite parameters). Terms and formulas contain only finitely many parameters:
finite (params-tm t)
finite (params-fm p)

Proof. By simple inductions. ◀

▶ Lemma 3 (Unused parameters). The denotation of an unused parameter does not affect
the semantics of either terms or formulas:

f /∈ params-tm t =⇒ (|E , F(f := x)|) t = (|E , F |) t
f /∈ params-fm p =⇒ [[E , F(f := x), G]] p = [[E , F , G]] p

Proof. By simple inductions. ◀

5.2 Instantiation
I will need to instantiate a universally quantified formula with a concrete term: to go from
∀ p to “p with t inserted for variable 0 and free variables in p adjusted.” I will denote this
formula by ⟨t/0⟩p. Note that when instantiating for n in ∀ p, we need to then instantiate for
n + 1 in p, since we enter the scope of another quantifier (the formula ∀x. ∀y. P (x, y) becomes
∀∀P (1, 0) with de Bruijn indices, so to instantiate for x we must actually replace variable 1).

There are two additional considerations. Consider first why we need to adjust the
free variables in p. Say that we are instantiating ∀ P [#0, #3] with the term t. When
evaluating ∀ P [#0, #3] under an environment E, the free variable 3 will be interpreted
as (E⟨0:x⟩) 3 = E 2. We expect that the interpretation of free variables under the same
environment does not change when we instantiate a quantifier. However, when evaluating
the naively instantiated formula P [t, #3], the free variable 3 will be evaluated as E 3, which
might be a different value than E 2. Therefore, we should decrement any free variables we
encounter during the instantiation. The result here should then be P [t, #2].

Second, it is important that any free variable in t remains free in ⟨t/0⟩p (i.e. that the
instantiation avoids capturing a free variable). With named variables we would ensure this
by renaming any bound variables in p that would conflict. By using de Bruijn indices we are
free from having to come up with fresh names for such an operation. Instead, we increment
every variable in t by one whenever we pass under a quantifier. Thus ⟨†f [#0]/0⟩(∀ (‡P )) =
∀ (⟨†f [#1]/1⟩(‡P )).

I call this last operation lifting the term:

primrec lift-tm (↑) where
↑(#n) = #(n+1 )
| ↑(†f ts) = †f (map ↑ ts)

While this terminology is common (cf. Nipkow [32], Berghofer [3]) it unfortunately
conflicts with the terminology in explicit substitutions (cf. Lescanne [28]) where lifting and
shifting have roughly opposite meanings compared to this paper.

With the above considerations in mind, we can now define instantiation on terms:
primrec inst-tm (⟨⟨- ′/-⟩⟩) where
⟨⟨s/m⟩⟩(#n) = (if n < m then #n else if n = m then s else #(n−1 ))
| ⟨⟨s/m⟩⟩(†f ts) = †f (map ⟨⟨s/m⟩⟩ ts)

The notation ⟨⟨s/m⟩⟩ “bundles” an instantiation of term s for variable m, ready to be
applied to a term. For formulas, the only interesting case is for the universal quantifier, where
we lift the term we are instantiating with and increment the variable we are instantiating for:

primrec inst-fm (⟨- ′/-⟩) where
⟨-/-⟩⊥ = ⊥
| ⟨s/m⟩(‡P ts) = ‡P (map ⟨⟨s/m⟩⟩ ts)
| ⟨s/m⟩(p −→ q) = ⟨s/m⟩p −→ ⟨s/m⟩q
| ⟨s/m⟩(∀ p) = ∀ (⟨↑s/m+1 ⟩p)

Despite the complexity of instantiation when using de Bruijn indices, it can be captured
in the three simple definitions above that involve little more than simple arithmetic.

A more standard name for ⟨t/n⟩p is substitution, but I prefer instantiation since it
potentially does more than simple syntactic substitution of term t for variable n: namely
lifts t and decrements variables in p.

The only results about instantiation that I need for the formalization are the following.

▶ Lemma 4 (Lifting and shifting). Lifting cancels out with shifting the environment at 0.
(|E⟨0 :x⟩, F |) (↑t) = (|E , F |) t

Proof. By structural induction. ◀

▶ Lemma 5 (Instantiation and shifting). Instantiating with a term at m is the same as shifting
the environment at m with the value denoted by that term.

(|E , F |) (⟨⟨s/m⟩⟩t) = (|E⟨m:(|E , F |) s⟩, F |) t
[[E , F , G]] (⟨t/m⟩p) = [[E⟨m:(|E , F |) t⟩, F , G]] p

Proof. By structural induction, using Lemma 4. ◀

5.3 Size
To prove the model existence theorem, I will need to do induction on formulas. However,
structural induction does not work, since in the case for ∀ p, the induction hypothesis must
be applied to the instance ⟨t/0⟩p, for some term t, rather than simply to p. This calls for
induction on the size of the formula. Unfortunately, the pre-defined size measure for our
datatype takes the size of terms into account and is therefore not invariant under instantiation.
The following definition suffices:

primrec size-fm where
size-fm ⊥ = 1
| size-fm (‡- -) = 1
| size-fm (p −→ q) = 1 + size-fm p + size-fm q
| size-fm (∀ p) = 1 + size-fm p

▶ Lemma 6 (Size). Instantiation preserves size.
size-fm (⟨t/m⟩p) = size-fm p

Proof. By structural induction. ◀

5.4 Propositional Semantics
Instead of picking a suitable set of propositional axioms, Smullyan [40], Barwise [1] and
others simply include all tautologies as one of their axioms. I follow their lead and need
a suitable way to express which formulas are tautologies. Smullyan [40, p. 51] extends his
notion of a Boolean valuation from propositional logic to the syntax of first-order logic by
treating quantified formulas as another sort of propositional symbols. A tautology is then a
formula that is true under all Boolean valuations.

The following definition uses the same principle, where G is a predicate denotation as
before and A is a special “universally quantified formula denotation.”

primrec boolean where
boolean - - ⊥ = False
| boolean G - (‡P ts) = G P ts
| boolean G A (p −→ q) = (boolean G A p −→ boolean G A q)
| boolean - A (∀ p) = A (∀ p)

The hyphens stand for ignored arguments. Compare this semantics to the first-order one:
it is indeed a Boolean valuation [40] of first-order logic. We can now take Smullyan’s notion
of tautology as definition:

abbreviation tautology p ≡ ∀G A. boolean G A p

Smullyan gives no details on his extension of Boolean valuations to first-order logic. The
way I set it up, with a separate denotation for the quantified formulas, it can be directly
related to the first-order semantics.

▶ Lemma 7 (Boolean semantics). The Boolean and first-order semantics coincide when G
matches the first-order predicate semantics and A is the first-order semantics itself.

boolean (λa. G a ◦ map (|E , F |)) [[E , F , G]] = [[E , F , G]]

Proof. By structural induction. ◀

▶ Lemma 8 (Tautologies). All tautologies are valid.
tautology p =⇒ [[E , F , G]] p

Proof. Since a tautology holds for any choice of G and A it holds in particular for those
that coincide with the first-order semantics (cf. Lemma 7). ◀

For reassurance, Isabelle easily verifies that not all first-order validities are propositional
tautologies (e.g. (∀x. P (x)) −→ P (a) is only the former):

proposition ∃ p. (∀E F G. [[E , F , G]] p) ∧ ¬ tautology p

5.5 The Inductively Defined Calculus
Finally, we are ready to define the calculus itself. I define it as an inductive predicate ⊢ that
holds exactly when a formula can be derived from the given axioms and rules. The previous
work has made the definition simple:

inductive Axiomatic (⊢ - [50 ] 50 ) where
TA: tautology p =⇒ ⊢ p
| IA: ⊢ ∀ p −→ ⟨t/0 ⟩p
| MP: ⊢ p −→ q =⇒ ⊢ p =⇒ ⊢ q
| GR: ⊢ q −→ ⟨⋆a/0 ⟩p =⇒ a /∈ params {p, q} =⇒ ⊢ q −→ ∀ p

The Tautology Axiom (TA) derives any tautology. The Instantiation Axiom (IA) states
that a quantified formula implies its instantiation with any term. The Modus Ponens (MP)
rule is stated as usual and lifts an implication between formulas to an implication between
derivations. Finally, the Generalization Rule (GR) works under assumptions q and generalizes
from an instance to a quantified formula, given that the witness (the constant) is fresh.

▶ Theorem 9 (Soundness). Any derivable formula is valid:
⊢ p =⇒ [[E , F , G]] p

Proof. By induction over the inductive definition of the axiomatic system for arbitrary
function denotation F .

All cases except for GR can be proven automatically, with the case for TA relying on
Lemma 8 about tautologies. In the GR case I apply the induction hypothesis not just once
at plain F but at F (a := x) for every element x of the domain:

have [[E , F(a := x), G]] (q −→ ⟨⋆a/0 ⟩p) for x

This is enough help for Isabelle to prove the case. ◀

▶ Corollary 10. Falsity cannot be derived:
¬ (⊢ ⊥)

5.5.1 Notation

For the proof of completeness I need to express that a formula can be derived from a set of
assumptions. Instead of building this notion into the definition of the proof system, I am
going to simulate it using chains of implications. The expression [p1, p2, . . . , pn]⇝ q expands
to p1 −→ p2 −→ . . . −→ pn −→ q. It is defined by recursion on the list of assumptions:

primrec imply (infixr ⇝ 56 ) where
([] ⇝ q) = q
| (p # ps ⇝ q) = (p −→ ps ⇝ q)

I then write ps ⊢ q to abbreviate ⊢ ps ⇝ q:
When I talk about assumptions in a derivation I will always mean a finite list of formulas.

5.6 Derived Formulas

Due to my semantic characterization of the Tautology Axiom, the automation in Isabelle
can easily prove that various propositional formulas (schemas) can be derived.

▶ Lemma 11 (Derivations). The S and K combinators, double negation elimination and
contraposition in both directions can all be derived:
⊢ (p −→ q −→ r) −→ (p −→ q) −→ p −→ r
⊢ q −→ p −→ q
⊢ ¬ ¬ p −→ p
⊢ (p −→ q) −→ ¬ q −→ ¬ p
⊢ (¬ q −→ ¬ p) −→ p −→ q

Proof. By the Tautology Axiom. ◀

5.6.1 Generalization Rule
My use of chains of implications is a disadvantage to the GR rule since it works on the
consequent but implication is right associative. Consider the following: we know that
ps ⊢ ⟨⋆a/0⟩p, for fresh a and want to use GR to derive ps ⊢ ∀ p. We can only do so if
ps consists of exactly one formula q, as ps ⊢ p is short for ⊢ ps ⇝ q. To circumvent this
restriction, I derive the following variant of the rule.

▶ Lemma 12 (GR’ rule). The following rule is derivable:

GR ′: ⊢ ¬ ⟨⋆a/0 ⟩p −→ q =⇒ a /∈ params {p, q} =⇒ ⊢ ¬ (∀ p) −→ q

Proof. Follows from the GR rule, modus ponens and the derivations in Lemma 11. ◀

Since this rule works on the left-hand side of the implication, the right-hand side can,
without issues, be an arbitrarily long chain of implications. Smullyan [40, p. 83] himself uses
this version of the rule in his System Q1’ (but for notational reasons).

An alternative is to start from the existential quantifier, ∃, as primitive, rather than ∀,
as the generalization rule for ∃ works on the left-hand side of the implication [40]. However,
it is less immediately clear why this rule for ∃ can be called a generalization rule.

5.6.2 Working with Assumptions
The following is an assortment of useful lemmas for working with assumptions.

▶ Lemma 13 (Assumptions). The following are derivable: modus ponens under assumptions,
derivation of any assumption, the deduction theorem in both directions, a cut rule, classical
reasoning and finally a structural rule encompassing weakening, contraction and exchange:

ps ⊢ p −→ q =⇒ ps ⊢ p =⇒ ps ⊢ q
p ∈ set ps =⇒ ps ⊢ p
ps ⊢ p −→ q =⇒ p # ps ⊢ q
p # ps ⊢ q =⇒ ps ⊢ p −→ q
p # ps ⊢ r =⇒ q # ps ⊢ p =⇒ q # ps ⊢ r
(¬ p) # ps ⊢ ⊥ =⇒ ps ⊢ p
ps ⊢ q =⇒ set ps ⊆ set ps ′ =⇒ ps ′ ⊢ q

Proof. By a mix of induction over the list of assumptions and propositional reasoning. ◀

6 Completeness

We are now ready to delve into the completeness proof itself. The plan is as follows. If we
cannot derive a formula p under any assumptions from X then we cannot derive falsity from
¬ p and any assumptions from X either. Sets like {¬ p} ∪ X are consistent with respect
to the proof system, as we cannot derive a contradiction from them. I formalize them in
Section 6.1. These sets are defined based on the proof system but we will use them to build
a model that contradicts the validity of p under X. For this purpose we must prove that
two important types of formulas preserve consistency: fresh witnesses of existential formulas
(Henkin witnesses) and instances of universal formulas.

Lindenbaum (according to Tarski [41]) showed how to extend a consistent set into a
maximal consistent set (MCS). Any proper superset of a maximal consistent set is inconsistent.
In particular this means that for any formula p, an MCS contains exactly p or ¬ p. Henkin [21],

showed the utility of adding the Henkin witnesses for existential formulas during Lindenbaum’s
construction. I formalize the construction and its consistency in Section 6.2 and prove that
the result is maximal in Section 6.3.

The addition of Henkin witnesses ensures that our MCSs are saturated. Section 6.4
outlines the benefits of ensuring this by construction.

Instead of building a model directly from a maximal consistent saturated set, I introduce
a standard layer of abstraction. In Section 6.5, I formalize the notion of a Hintikka set [40]
using three simple conditions and prove a model existence theorem: given a Hintikka set H,
I build a model from a Herbrand structure [10, 22] that satisfies exactly the formulas in H. I
then prove that maximal consistent saturated sets are Hintikka sets.

In Section 6.6, I put all the pieces together. The model existence theorem gives us a
model for ¬ p and all of X. Therefore, if p is in fact valid under assumptions from X, then
it must be derivable or we have a contradiction.

6.1 Consistent Sets
The definition of consistency is straightforward. The set of formulas S is consistent when
there is no list of assumptions S’, coming from S, that can be used to derive falsity:

definition consistent S ≡ ∄S ′. set S ′ ⊆ S ∧ S ′ ⊢ ⊥

The following lemma will be useful.

▶ Lemma 14 (Inconsistent addition). Assume that S is consistent on its own but becomes in-
consistent with the addition of a formula p. Then there exists a concrete list of assumptions S′,
coming from S, such that p # S ′ ⊢ ⊥:

assumes consistent S and ¬ consistent ({p} ∪ S)
obtains S ′ where set S ′ ⊆ S and p # S ′ ⊢ ⊥

Proof. It follows from consistency and the structural lemma for assumptions (Lemma 13). ◀

It is important to prove that two types of formulas preserve consistency. The first type is
fresh witnesses for existential formulas.

▶ Lemma 15 (Consistency of fresh witnesses). If a consistent set contains an existential
formula ¬ (∀ p) then adding a witness ¬ ⟨⋆a/0⟩p, for a fresh a, preserves consistency:

assumes consistent S and ¬ (∀ p) ∈ S and a /∈ params S
shows consistent ({¬ ⟨⋆a/0 ⟩p} ∪ S)

Proof. We need to show that there is no finite subset from which we can derive falsity, so
assume that indeed there is one. From Lemma 14 we can name the problematic assumptions:

then obtain S ′ where set S ′ ⊆ S and (¬ ⟨⋆a/0 ⟩p) # S ′ ⊢ ⊥

After showing that the side conditions are fulfilled, we can apply the GR’ rule:

then have ¬ (∀ p) # S ′ ⊢ ⊥

But every assumption is now in S, which we assumed to be consistent, so we have reached
the desired contradiction and proved the lemma. ◀

We shall also need that instantiating a universally quantified formula preserves consistency.

▶ Lemma 16 (Consistency of instantiation). If a consistent set contains a universal formula ∀ p

then adding an instance ⟨t/0⟩p, for any term t, preserves consistency:

assumes consistent S and ∀ p ∈ S
shows consistent ({⟨t/0 ⟩p} ∪ S)

Proof. The proof proceeds as before and we start by naming the problematic assumptions
from an assumed inconsistency (Lemma 14):

then obtain S ′ where set S ′ ⊆ S and ⟨t/0 ⟩p # S ′ ⊢ ⊥

This time we make use of the Instantiation Axiom, instantiated to p and t:

moreover have ⊢ ∀ p −→ ⟨t/0 ⟩p

With the deduction theorem, the cut rule and the structural lemma (Lemma 13), we can
apply this implication to weaken the derivation of falsity:

ultimately have ∀ p # S ′ ⊢ ⊥

But again, these assumptions are all in S, which we assumed to be consistent, so this is a
contradiction and adding the instance must also be consistent. ◀

6.2 Lindenbaum Extension
We turn now to a central construction. Note first that if the sets of variable, function and
predicate symbols are countable, so too are the sets of terms and formulas (formalized in
Section 6.6). Thus, we can enumerate the formulas as p0, p1, . . . and so on. Starting from a
consistent set S0, which leaves infinitely many parameters unused, we then build a sequence
of consistent sets in the following way. Given Sn, construct Sn+1 as:

Sn+1 =
{

w(∗, pn) ∪ {pn} ∪ Sn if {pn} ∪ Sn is consistent
Sn otherwise

where ∗ is the set of parameters in {pn} ∪ Sn.
The function w returns a singleton set with a fresh witness when pn is an existential

formula and the empty set otherwise. Usually, the availability of such fresh witnesses is
guaranteed by extending the set of function symbols. I assume instead that the set of function
symbols is infinite from the start and that S0 leaves infinitely many parameters unused. I
pass the parameters of {pn} ∪ Sn to w. It can then pick a parameter that has not been used
already. This is simpler than dealing with two sorts of function symbols.

In the Isabelle formalization, the enumeration of formulas is represented by a (surjective)
function f from the set of natural numbers to the set of formulas (cf. Section 6.6). The
expression extend S f n constructs the set Sn starting from S0 = S:

primrec extend where
extend S f 0 = S
| extend S f (Suc n) =

(let Sn = extend S f n in
if consistent ({f n} ∪ Sn)
then witness (params ({f n} ∪ Sn)) (f n) ∪ {f n} ∪ Sn
else Sn)

The function witness is simple:

fun witness where
witness used (¬ (∀ p)) = {¬ ⟨⋆(SOME a. a /∈ used)/0 ⟩p}
| witness - - = {}

Its definition uses Hilbert’s choice operator to pick a fresh parameter.
The maximal consistent set is given by taking the union of this sequence of sets:

⋃
n∈N Sn.

In Isabelle, it becomes:
definition Extend S f ≡

⋃
n. extend S f n

The following lemmas are needed later.

▶ Lemma 17 (Lindenbaum bounds). The starting set is included in its maximal extension
and each set in the constructed sequence bounds the previous sets:

S ⊆ Extend S f
(
⋃

n ≤ m. extend S f n) = extend S f m

Proof. By definition and by induction on m, respectively. ◀

▶ Lemma 18 (Lindenbaum parameters). A witness includes only finitely many parameters
and each set Sn contains finitely many more parameters than the starting set S0:

finite (params (witness used p))
finite (params (extend S f n) − params S)

Proof. Since p contains finitely many parameters and by induction on n, respectively. ◀

6.2.1 Consistency
The consistency of each constructed set Sn is apparent.

▶ Lemma 19 (Consistency of Sn). When starting from a consistent S0 with infinitely many
unused parameters, any constructed Sn is consistent:

assumes consistent S and infinite (UNIV − params S)
shows consistent (extend S f n)

Proof. By induction on n. The consistency of adding the witness follows from Lemma 15.
The only complication is to prove that there are indeed always fresh parameters available
and therefore that the parameter given by Hilbert’s choice operator is usable, but this follows
from Lemma 18. ◀

The consistency of the union
⋃

n Sn is more interesting.

▶ Lemma 20 (Consistency of
⋃

n Sn). The maximal extension of a consistent set S with
infinitely many unused parameters is consistent:

assumes consistent S and infinite (UNIV − params S)
shows consistent (Extend S f )

Proof. Assume towards a contradiction that we can derive falsity from some finite subset:
then obtain S ′ where S ′ ⊢ ⊥ and set S ′ ⊆ Extend S f

Since this subset is finite, it must be a subset of some initial segment of the union:
then obtain m where set S ′ ⊆ (

⋃
n ≤ m. extend S f n)

But, by Lemma 17, each such segment is bounded by its last element:
then have set S ′ ⊆ extend S f m

And since we have already shown the consistency of each Sn (Lemma 19), we reach our
desired contradiction. ◀

6.3 Maximal Sets
A maximal set is inconsistent under any proper extension:

definition maximal S ≡ ∀ p. p /∈ S −→ ¬ consistent ({p} ∪ S)

Maximal consistent sets are truly maximal:

▶ Lemma 21 (Maximality of Maximal Consistent Sets). If S is a maximal consistent set, then
for every formula p, p ∈ S if and only if ¬ p /∈ S.

assumes consistent S and maximal S
shows p ∈ S ←→ (¬ p) /∈ S

Proof. The left-to-right direction follows from consistency alone and the right-to-left direction
follows from consistency and maximality. ◀

That the Lindenbaum extension results in a maximal set is very easy to see.

▶ Lemma 22 (Maximality of
⋃

n Sn). Given a surjective enumeration f ,
⋃

n Sn is maximal:

assumes surj f
shows maximal (Extend S f )

Proof. Assume towards a contradiction that some formula p is not included even though its
inclusion preserves consistency:

assume p /∈ Extend S f and consistent ({p} ∪ Extend S f )

Say that p is formula number k in the enumeration. Since p is not in the result, it must
be inconsistent with Sk:

then have ¬ consistent ({p} ∪ extend S f k)

And this set is a subset of the final result:

moreover have {p} ∪ extend S f k ⊆ {p} ∪ Extend S f

Ultimately, this contradicts the assumption that adding p preserves consistency. ◀

6.4 Saturation
We shall need saturation to show that our constructed sets are Hintikka sets:

definition saturated S ≡ ∀ p. ¬ (∀ p) ∈ S −→ (∃ a. (¬ ⟨⋆a/0 ⟩p) ∈ S)

So, in a saturated set there is a corresponding Henkin witness for each existential formula.

▶ Lemma 23 (Saturation of
⋃

n Sn). A consistent Lindenbaum extension is saturated:

assumes consistent (Extend S f ) and surj f
shows saturated (Extend S f )

Proof. Guaranteed by construction. ◀

If we only constructed our set to be maximal consistent and tried to show that it was also
saturated, we would run into trouble [40, p. 96]. First, given an arbitrary maximal consistent
set S, it might be that a Henkin witness is missing because S includes every parameter
available and every reuse of a parameter results in an inconsistency. Second, we might be
unlucky and enumerate the negation of every suitable witness before enumerating the witness
itself: we might always add the negation and never the witness. Following Henkin [21], I
ensure saturation by adding the Henkin witnesses together with the existential formulas.

6.5 Hintikka Sets
Instead of showing the model existence theorem directly for maximal consistent saturated
sets, it will be cleaner to show that Hintikka sets induce a model for their formulas and that
our sets are in fact Hintikka sets.

The following definition characterizes a Hintikka set H over our syntax:

locale Hintikka =
fixes H :: ( ′f , ′p) fm set
assumes

FlsH : ⊥ /∈ H and
ImpH : (p −→ q) ∈ H ←→ (p ∈ H −→ q ∈ H ) and
UniH : (∀ p ∈ H ) ←→ (∀ t. ⟨t/0 ⟩p ∈ H )

Hintikka sets are sets that are saturated downwards [40, p. 27] and induce a model for the
formulas in them. Since the set should induce a model, ⊥ should never be present (FlsH ).
Following Forster et al. [11, Lemma 11], I enforce that the set respects both implication
(ImpH ) and universal quantification (UniH ): a formula is in the Hintikka set if and only if
the “evidence” for that formula is also present. Here, evidence is to be understood in terms
of the Herbrand model given below.

6.5.1 Model Existence
The model induced by a Hintikka set H is very simple. It consists of a Herbrand structure [10]
and a predicate denotation based on H itself:
Domain Herbrand universe: the universe of terms.
Function denotation The constructor †, i.e. every function symbol evaluates to itself.
Predicate denotation Predicate P is true for terms ts exactly when ‡P ts ∈ H.

Like in the work by Herbelin and Ilik [22], but unlike for instance the formalizations by
Berghofer [3] and Forster et al. [11], the Herbrand universe includes all terms, not just those
with no variables. I never formalize what it means for a formula to be closed. The Herbrand
structure famously evaluates any term without variables to itself [10]. Or in this case:

▶ Lemma 24 (Herbrand semantics). Under any Herbrand structure and the specific environ-
ment #, every term evaluates to itself:

(|#, †|) t = t

Proof. By structural induction. ◀

I reuse the notation for semantics and abbreviate the model induced by H as [[H ]]:

abbreviation hmodel ( [[-]] ) where [[H ]] ≡ [[#, †, λP ts. ‡P ts ∈ H ]]

We now reach the model existence theorem.

▶ Theorem 25 (Model existence). When H is a Hintikka set, [[H ]] satisfies exactly the
formulas in H.

assumes Hintikka H
shows p ∈ H ←→ [[H ]] p

Proof. By well-founded induction on the size of the formula as given by size-fm. Thus the
induction hypothesis applies to any formula that is smaller by this measure, i.e. to subformulas
and to instances of universally quantified formulas (cf. Lemma 6). These are exactly the

formulas that appear in the Hintikka conditions. The proof proceeds by considering each
type of formula. Since there is a Hintikka condition for every type, which corresponds exactly
to the semantics of the induced model, Isabelle automatically proves each case. For instance,
a universal formula p is in the Hintikka set iff every instance ⟨t/0⟩p is in the Hintikka set
(UniH ) iff every instance ⟨t/0⟩p holds in the induced model (by the induction hypothesis). ◀

6.5.2 Saturated MCSs are Hintikka Sets
Consider first the following correspondence between derivability and MCSs.

▶ Lemma 26 (Derivability and MCSs). A formula p is derivable from an MCS S iff p is in S:

assumes consistent S and maximal S
shows (∃ ps. set ps ⊆ S ∧ ps ⊢ p) ←→ p ∈ S

Proof. The left to right direction follows from the maximality of MCSs. The right to left
direction follows trivially from the derivability of any assumption (Lemma 13). ◀

I now show that maximal consistent saturated sets are Hintikka sets.

▶ Lemma 27 (Saturated MCSs are Hintikka sets). If the set H is consistent, maximal and
saturated, it is a Hintikka set:

assumes consistent H and maximal H and saturated H
shows Hintikka H

Proof. We need to prove each case of the Hintikka definition. Take first the FlsH case:

show ⊥ /∈ H

We need to show that falsity does not appear in our set. This follows directly from
Lemma 26 and the assumed consistency of H.

Consider next the ImpH case:

show (p −→ q) ∈ H ←→ (p ∈ H −→ q ∈ H )

From left to right, by using Lemma 26 this simply becomes modus ponens: if both p −→ q

and p are derivable from H then q must be derivable from H. The right to left direction is
similar. It relies on Lemma 26, contraposition and Lemma 21: that exactly one of a formula
and its negation is present in an MCS.

Consider next the UniH case:

show (∀ p ∈ H ) ←→ (∀ t. ⟨t/0 ⟩p ∈ H )

One direction follows directly from consistency of instantiation (Lemma 16) and the
maximality of H. The other direction follows from saturation (and Lemma 21). ◀

6.6 Completeness Theorem
Isabelle can automatically prove the countability of our syntax:

instance tm :: (countable) countable
instance fm :: (countable, countable) countable

These commands provide instances of the surjective function from-nat that takes natural
numbers and returns terms and formulas, respectively. I state the main theorem as follows.

▶ Theorem 28 (Completeness). Assume that formula p is valid under assumptions X and
that X leaves infinitely many parameters unused. Then we can derive p from X.

fixes p :: ( ′f :: countable, ′p :: countable) fm
assumes ∀ (E :: - ⇒ ′f tm) F G. (∀ q ∈ X . [[E , F , G]] q) −→ [[E , F , G]] p

and infinite (UNIV − params X)
shows ∃ ps. set ps ⊆ X ∧ ps ⊢ p

Proof. By contradiction:
assume ∄ ps. set ps ⊆ X ∧ ps ⊢ p
then have ∗: ∄ ps. set ps ⊆ X ∧ ((¬ p) # ps ⊢ ⊥)

If no such list of assumptions exists, then (by classical reasoning on the object level) there
is also no list that allows us to derive falsity when assuming ¬ p.

I introduce some local abbreviations ?S and ?H (where ? is required by Isabelle):
let ?S = {¬ p} ∪ X
let ?H = Extend ?S from-nat

It is easy to see from ∗ above that ?S must be consistent and the extension ?H is therefore
maximal consistent (Lemmas 20, 22):

have consistent ?S
moreover have infinite (UNIV − params ?S)
ultimately have consistent ?H and maximal ?H

?H is saturated (Lemma 23) and Hintikka (Lemma 27):
moreover from this have saturated ?H
ultimately have Hintikka ?H

The model induced by ?H satisfies any formula in ?H (Theorem 25), including the
starting set ?S (Lemma 17):

have [[?H ]] p if p ∈ ?S for p
then have [[?H ]] (¬ p) and ∀ q ∈ X . [[?H ]] q

But this includes all formulas in X so by the assumed validity, [[H ]] must also satisfy p
and we reach our contradiction:

moreover from this have [[?H ]] p
ultimately show False

The proof system is complete. ◀

The following abbreviation of validity in a specific Herbrand universe, with countably
infinite function and predicate symbols, makes the result simpler to state:

abbreviation valid :: (nat, nat) fm ⇒ bool where
valid p ≡ ∀ (E :: nat ⇒ nat tm) F G. [[E , F , G]] p

I fix the function and predicate symbols to be natural numbers but any countably infinite
type works. One thing to note is that I only assume validity in one domain (the Herbrand
universe), as I cannot quantify over the type I use to represent the domain. This is, however,
a weaker assumption than assuming validity in all domains as is usually done.
▶ Theorem 29 (Soundness and completeness). Exactly the valid formulas are derivable:

theorem main: valid p ←→ (⊢ p)

Proof. By Theorems 9, 28. ◀

Only the definitions in Sections 3, 4 and Sections 5.1 to 5.5 must be inspected to trust
the result. The definitions in this Section are only used for the proof.

TYPES 2021

7 Discussion

There are many choices to make in a formalization like this one. I choose to work with
de Bruijn indices rather than named variables or Nominal Isabelle [42], which provides
automation for this situation. While this choice makes it more complicated to explain the
formalizations of e.g. semantics and quantifier instantiation, it makes the formalization
self-contained. I hope to have demonstrated that the definitions themselves are simple, the
functions are short and only a few simple lemmas are needed about them.

Recall the GR rule which is used in Lemma 15 to justify the consistency of fresh witnesses:

GR: ⊢ q −→ ⟨⋆a/0 ⟩p =⇒ a /∈ params {p, q} =⇒ ⊢ q −→ ∀ p

Since I use de Bruijn indices, this could also be formalized without the use of a parameter a

by lifting q, in the sense of ↑ , to ensure that variable 0 in p is safe to generalize directly:

⊢ ↑q → p =⇒ ⊢ q → ∀ p

However, we would then need to ensure that the entire set S ′ in Lemma 15 is lifted in order
to apply the rule. With the present GR rule, we simply ensure that a is chosen to be fresh.
It would be interesting to try Laurent’s anti-locally nameless approach to quantifiers [27]
and see whether this would yield a simpler formalization.

Another choice has been to simulate assumptions in derivations by a chain of implications.
This trick applies directly to a one-sided calculus and makes it a lot simpler to work with,
especially with some custom notation. It works especially well with Smullyan’s System Q1
where the generalization rule (GR) works under an implication. The semantic characterization
of the tautology axiom, which works well with Isabelle’s proof automation, makes it even
smoother since propositional reasoning becomes a non-issue.

One challenge was the realization that the variant GR’ is more suitable than GR. Isabelle
cannot tell us something like this, nor is the proof automation powerful enough to derive the
rule automatically. The insight comes from experimenting with the formalization and proofs.

Some of these issues could also be resolved by starting from a natural deduction system
rather than Smullyan’s Hilbert system. Natural deduction systems have a context built in,
where I must simulate it with implications, and more natural rules for the connectives, which
could be used instead of the semantic characterization of tautologies. It remains future work
to adapt the formalization to this setting and review the potential benefits.

At this point in time there is a large body of formalizations to draw on. I am inspired by
Berghofer’s formalization [3] of the completeness of natural deduction for first-order logic.
Berghofer also formalizes Lindenbaum’s construction and my definition is close to his. My
formalization of Hintikka sets and the model existence theorem, however, is both shorter (due
to Forster et al. [11]) and, unlike Berghofer’s, works directly for open formulas (cf. Herbelin
and Ilik [22]). As such, even though some notion has already been formalized, it can be
beneficial to revisit it.

8 Conclusion and Future Work

I have used techniques from computer science like de Bruijn indices and functional program-
ming to work in the meta-logic of the proof assistant Isabelle/HOL. Here, I have formalized
the syntax and semantics of first-order logic and defined a simple axiomatic proof system for
it. This definition has included careful considerations of the interplay between syntax and
semantics, a semantic characterization of tautologies suitable for formalization and notational
tricks like the use of implications to simulate assumptions.

I have then carried out a completeness proof for the Hilbert system in the style of Henkin,
and using ideas from Lindenbaum, Hintikka and Herbrand along the way. The proof is
direct: use Lindenbaum’s construction to extend a consistent set to a maximal consistent
set, add Henkin witnesses of existential formulas during this construction, notice that the
result is a Hintikka set and build a model in the Herbrand universe. Section 2 demonstrated
the usefulness of this style in the formalization of other logics and proof systems. My
formalization may serve as starting point for such endeavors: researchers can modify the
existing definitions and proofs rather than start from scratch. Isabelle/HOL ensures that
such modifications are correct and can help fill in gaps in the proofs when they arise. This
provides an entry point to formalizing such a completeness proof.

In the future, however, I want to abstract this proof along several dimensions. First, the
entire construction outlined above could potentially be given in the abstract and instantiated
with a concrete proof system, witnessing function, notion of saturation, etc. Then it might
be shared among the several formalizations of this method, and potential new ones. Popescu
and Traytel [36] have already developed some syntax-independent logical infrastructure in
their formal verification of an abstract account of Gödel’s incompleteness theorems. This
future work could potentially build on theirs, extending it with the model existence theorem
and more. Second, Smullyan gives many constructions in his uniform notation that abstracts
over the concrete choice of syntax. I would like to abstract this formalization in a similar way:
witnesses could be added for “δ-formulas”, which might happen to be of the form ¬ (∀ p)
like here or maybe of the form ♢ p as seen in hybrid logic [26].

I also want to extend the syntax, semantics, proof system and completeness proof to
first-order logic with equality. I already handle function symbols, unlike Smullyan, but to get
on par with Barwise, equality needs to be considered too. The Henkin style should scale well
for this extension. The current formalization does, however, have the benefit of outlining
the fundamental ideas of the completeness proof without too many auxiliary considerations.
This is an advantage for adapting it to other logics.

I hope this formalization will serve as inspiration, and perhaps as a starting point, for
further formalizations of logic.

